

# **Programming Languages – Describing Syntax and Semantics**

---

Jongwoo Lim

# Introduction

---

- **Syntax:** the form or structure of the expressions, statements, and program units
- **Semantics:** the meaning of the expressions, statements, and program units
  - Syntax and semantics provide a language's definition
    - Users of a language definition
    - Other language designers
    - Implementers
    - Programmers (the users of the language)

# The General Problem of Describing Syntax: Terminology

---

- A **language** is a set of sentences.
- A **sentence** is a string of characters over some alphabet.
  - I like Programming Languages.  
`index = 2 * count + 17;`
- **Lexeme** is the lowest level syntactic unit of a language.
  - I, like, Programming, Languages, .  
`index, =, 2, *, ;`
- **Token** is a category of lexemes.
  - pronoun, verb, noun, symbol\_period, ...  
`identifier, equal_sign, int_literal, ...`

# Formal Definition of Languages

---

- **Recognizers**

- Is the given sentence in the language?
- A recognition device reads input strings over the alphabet of the language and decides whether the input strings belong to the language.
- Example: syntax analysis part of a compiler.
- Detailed discussion of syntax analysis appears in Chapter 4.

- **Generators**

- A device that generates sentences of a language.
- One can determine if the syntax of a particular sentence is syntactically correct by comparing it to the structure of the generator.

# BNF and Context-Free Grammars

---

- **Context-Free Grammars**

- Developed by Noam Chomsky in the mid-1950s.
- Language generators, meant to describe the syntax of natural languages.
- Define a class of languages called context-free languages.
- Block structure:

John, whose blue car was in the garage, walked to the store.

(John, ((whose blue car) (was (in the garage))), (walked (to (the store)))).

# BNF and Context-Free Grammars

---

- **Backus-Naur Form (1959)**
  - Invented by John Backus to describe Algol 58.
  - BNF is equivalent to context-free grammars.
  - BNF is a metalanguage for programming languages.

# BNF and Context-Free Grammars

---

- In BNF, **abstractions** are used to represent classes of syntactic structures – they act like syntactic variables.
  - Also called **nonterminal symbols**, or just **nonterminals**.
  - Nonterminals are often enclosed in angle brackets.  
`<identifier>`, `<equal_sign>`, `<int_literal>`
- **Terminals** are lexemes or tokens.
- A **rule** has a left-hand side (LHS) and a right-hand side (RHS).
  - A left-hand side (LHS) is a nonterminal.
  - A right-hand side (RHS) is a string of terminals and/or nonterminals.

`<assign>` → `<var> = <expression>`

# BNF Fundamentals

---

- A nonterminal can have more than one RHS.

```
<if_stmt> → if ( <logic_expr> ) <stmt>  
          | if ( <logic_expr> ) <stmt> else <stmt>
```

- Examples:

```
if (i == 0) a = b + 1;  
if (a > 0.0) positive = true; else positive = false;
```

- Recursive definition:

```
<ident_list> → <ident>  
             | <ident> , <ident_list>
```

- Examples:

```
1           (O)  
1,2,3,4     (O)  
1,2 3,4     (X)
```



# BNF Fundamentals

---

- **Grammar:** a finite non-empty set of rules
  - The sentences are generated through applications of the rules, beginning with the **start symbol** (a nonterminal).

```
<program> → begin <stmt_list> end  
<stmt_list> → <stmt>  
               | <stmt> ; <stmt_list>  
<stmt> → <var> = <expression>  
<var> → A | B | C  
<expression> → <var> + <var>  
               | <var> - <var>  
               | <var>
```

# Derivation

---

- Derivation:
  - Repeated application of rules, starting with the start symbol and ending with a sentence (all terminal symbols).
  - **Sentential form**: string of symbols in a derivation.
    - A sentence is a sentential form that has only terminal symbols.
  - **Leftmost derivation**:  
the leftmost nonterminal in each sentential form is expanded.
  - A derivation may be neither leftmost nor rightmost.

# An Example Derivation

```
<program>    => begin <stmt_list> end
              => begin <stmt> ; <stmt_list> end
              => begin <var> = <expression> ; <stmt_list> end
              => begin A = <expression> ; <stmt_list> end
              => begin A = <var> + <var> ; <stmt_list> end
              => begin A = B + <var> ; <stmt_list> end
              => begin A = B + C ; <stmt_list> end
              => begin A = B + C ; <stmt> end
              => begin A = B + C ; <var> = <expression> end
              => begin A = B + C ; B = <expression> end
              => begin A = B + C ; B = <var> end
              => begin A = B + C ; B = C end
```

## Example Small Language

```
<program> → begin <stmt_list> end
<stmt_list> → <stmt>
              | <stmt> ; <stmt_list>
<stmt> → <var> = <expression>
<var> → A | B | C
<expression> → <var> | <var> + <var>
              | <var> - <var>
```

# Another Example: Simple Assignment

- Example statement:

$A = B * ( A + C )$

- Leftmost derivation:

```
<assign>  => <id> = <expr>
           => A = <expr>
           => A = <id> * <expr>
           => A = B * <expr>
           => A = B * ( <expr> )
           => A = B * ( <id> + <expr> )
           => A = B * ( A + <expr> )
           => A = B * ( A + <id> )
           => A = B * ( A + C )
```

## Simple Assignment

```
<assign> → <id> = <expr>
<id> → A | B | C
<expr> → <id> + <expr>
        | <id> * <expr>
        | ( <expr> )
        | <id>
```

# Parse Tree

- A hierarchical representation of a derivation.

## Simple Assignment

$\langle \text{assign} \rangle \rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle$

$\langle \text{id} \rangle \rightarrow A \mid B \mid C$

$\langle \text{expr} \rangle \rightarrow \langle \text{id} \rangle + \langle \text{expr} \rangle$

$\mid \langle \text{id} \rangle * \langle \text{expr} \rangle$

$\mid ( \langle \text{expr} \rangle )$

$\mid \langle \text{id} \rangle$

$A = B * ( A + C )$

$\langle \text{assign} \rangle \Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle$

$\Rightarrow A = \langle \text{expr} \rangle$

$\Rightarrow A = \langle \text{id} \rangle * \langle \text{expr} \rangle$

$\Rightarrow A = B * \langle \text{expr} \rangle$


$\Rightarrow A = B * ( \langle \text{expr} \rangle )$

$\Rightarrow A = B * ( \langle \text{id} \rangle + \langle \text{expr} \rangle )$

$\Rightarrow A = B * ( A + \langle \text{expr} \rangle )$

$\Rightarrow A = B * ( A + \langle \text{id} \rangle )$

$\Rightarrow A = B * ( A + C )$



# Simple Assignment

- Can the grammar accept the following equation?

$$A = ( A + C ) * B$$

## Simple Assignment

```
<assign> → <id> = <expr>
<id> → A | B | C
<expr> → <id> + <expr>
        | <id> * <expr>
        | ( <expr> )
        | <id>
```

## Modified Simple Assignment

```
<assign> → <id> = <expr>
<id> → A | B | C
<expr> → <expr> + <expr>
        | <expr> * <expr>
        | ( <expr> )
        | <id>
```

# Parse Tree with Modified Grammar

## Modified Simple Assignment

$\langle \text{assign} \rangle \rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle$

$\langle \text{id} \rangle \rightarrow A \mid B \mid C$

$\langle \text{expr} \rangle \rightarrow \langle \text{expr} \rangle + \langle \text{expr} \rangle$

$\mid \langle \text{expr} \rangle * \langle \text{expr} \rangle$

$\mid ( \langle \text{expr} \rangle )$

$\mid \langle \text{id} \rangle$

$A = ( A + C ) * B$

$\langle \text{assign} \rangle \Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle$

$\Rightarrow A = \langle \text{expr} \rangle$

$\Rightarrow A = \langle \text{expr} \rangle * \langle \text{expr} \rangle$

$\Rightarrow A = ( \langle \text{expr} \rangle ) * \langle \text{expr} \rangle$

$\Rightarrow A = ( \langle \text{expr} \rangle + \langle \text{expr} \rangle ) * \langle \text{expr} \rangle$

$\Rightarrow A = ( \langle \text{id} \rangle + \langle \text{expr} \rangle ) * \langle \text{expr} \rangle$


$\Rightarrow A = ( A + \langle \text{expr} \rangle ) * \langle \text{expr} \rangle$

$\Rightarrow A = ( A + \langle \text{id} \rangle ) * \langle \text{expr} \rangle$

$\Rightarrow A = ( A + C ) * \langle \text{expr} \rangle$

$\Rightarrow A = ( A + C ) * \langle \text{id} \rangle$

$\Rightarrow A = ( A + C ) * B$



# Ambiguous Grammar

- Example statement:

A = B + C \* A

Modified Simple Assignment : Ambiguous

$\langle \text{assign} \rangle \rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle$



$\langle \text{id} \rangle \rightarrow A \mid B \mid C$

$\langle \text{expr} \rangle \rightarrow \langle \text{expr} \rangle + \langle \text{expr} \rangle$

$\quad \mid \langle \text{expr} \rangle * \langle \text{expr} \rangle$

$\quad \mid ( \langle \text{expr} \rangle )$

$\quad \mid \langle \text{id} \rangle$






# Simple Assignment Revisited

- The precedence order of operators is not usual.

$A = B + A * C$

$A = B * A + C$



## Simple Assignment

$\langle\text{assign}\rangle \rightarrow \langle\text{id}\rangle = \langle\text{expr}\rangle$


$\langle\text{id}\rangle \rightarrow A \mid B \mid C$

$\langle\text{expr}\rangle \rightarrow \langle\text{id}\rangle + \langle\text{expr}\rangle$

$\quad \mid \langle\text{id}\rangle * \langle\text{expr}\rangle$

$\quad \mid ( \langle\text{expr}\rangle )$

$\quad \mid \langle\text{id}\rangle$



# An Unambiguous Expression Grammar

- Ambiguity can be resolved by indicating precedence levels of the operators.

## Unambiguous Simple Assignment

```
<assign> → <id> = <expr>
<id> → A | B | C
<expr> → <expr> + <term>
        | <term>
<term> → <term> * <factor>
        | <factor>
<factor> → ( <expr> )
          | <id>
```

# Parse Tree with Unambiguous Grammar

## Unambiguous Simple Assignment

$\langle \text{assign} \rangle \rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle$

$\langle \text{id} \rangle \rightarrow A \mid B \mid C$

$\langle \text{expr} \rangle \rightarrow \langle \text{expr} \rangle + \langle \text{term} \rangle \mid \langle \text{term} \rangle$

$\langle \text{term} \rangle \rightarrow \langle \text{term} \rangle * \langle \text{factor} \rangle \mid \langle \text{factor} \rangle$

$\langle \text{factor} \rangle \rightarrow ( \langle \text{expr} \rangle ) \mid \langle \text{id} \rangle$

$A = B + C * A$

$\langle \text{assign} \rangle \Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle$

$\Rightarrow A = \langle \text{expr} \rangle$

$\Rightarrow A = \langle \text{expr} \rangle + \langle \text{term} \rangle$

$\Rightarrow A = \langle \text{term} \rangle + \langle \text{term} \rangle$

$\Rightarrow A = \langle \text{factor} \rangle + \langle \text{term} \rangle$

$\Rightarrow A = \langle \text{id} \rangle + \langle \text{term} \rangle$

$\Rightarrow A = B + \langle \text{term} \rangle$

$\Rightarrow A = B + \langle \text{term} \rangle * \langle \text{factor} \rangle$


$\Rightarrow A = B + \langle \text{factor} \rangle * \langle \text{factor} \rangle$

$\Rightarrow A = B + \langle \text{id} \rangle * \langle \text{factor} \rangle$

$\Rightarrow A = B + C * \langle \text{factor} \rangle$

$\Rightarrow A = B + C * \langle \text{id} \rangle$

$\Rightarrow A = B + C * A$



← Leftmost derivation

# Parse Tree from Rightmost Derivation

## Unambiguous Simple Assignment

$\langle \text{assign} \rangle \rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle$

$\langle \text{id} \rangle \rightarrow A \mid B \mid C$

$\langle \text{expr} \rangle \rightarrow \langle \text{expr} \rangle + \langle \text{term} \rangle \mid \langle \text{term} \rangle$

$\langle \text{term} \rangle \rightarrow \langle \text{term} \rangle * \langle \text{factor} \rangle \mid \langle \text{factor} \rangle$

$\langle \text{factor} \rangle \rightarrow ( \langle \text{expr} \rangle ) \mid \langle \text{id} \rangle$

$A = B + C * A$

$\langle \text{assign} \rangle \Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle + \langle \text{term} \rangle$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle + \langle \text{term} \rangle * \langle \text{factor} \rangle$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle + \langle \text{term} \rangle * \langle \text{id} \rangle$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle + \langle \text{term} \rangle * A$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle + \langle \text{factor} \rangle * A$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle + \langle \text{id} \rangle * A$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle + C * A$


$\Rightarrow \langle \text{id} \rangle = \langle \text{term} \rangle + C * A$

$\Rightarrow \langle \text{id} \rangle = \langle \text{factor} \rangle + C * A$

$\Rightarrow \langle \text{id} \rangle = \langle \text{id} \rangle + C * A$

$\Rightarrow \langle \text{id} \rangle = B + C * A$

$\Rightarrow A = B + C * A$



Note: the parse trees are same.

# Associativity of Operators

---

- Operators with the same precedence

- Example:

$$A = A + B + C$$

$$(A + B) + C == A + (B + C)$$

- Can be problematic in floating-point operations.

- Example:  $(-1 + 1) + 1e-200 \neq -1 + (1 + 1e-200)$

- Left- and right-recursion:

$\langle \text{expr} \rangle \rightarrow \langle \text{expr} \rangle + \langle \text{term} \rangle$  vs.

$\langle \text{factor} \rangle \rightarrow \langle \text{exp} \rangle ** \langle \text{factor} \rangle$

# Unambiguous Grammar for if-then-else

- Example statement:



**if** <logic\_expr> **then** **if** <logic\_expr> **then** <stmt> **else** <stmt>

Ambiguous if-then-else

<stmt> → <if\_stmt> | ...

<if\_stmt> → **if** <logic\_expr> **then** <stmt>

          | **if** <logic\_expr> **then** <stmt> **else** <stmt>



# Unambiguous Grammar for if-then-else

- Match else to the nearest then.

Unambiguous if-then-else

`<stmt> → <matched> | <unmatched> | ...`

`<matched> → if <logic_expr> then <matched> else <matched>`  
`| <any_non_if_statement>`

`<unmatched> → if <logic_expr> then <stmt>`  
`| if <logic_expr> then <matched> else <unmatched>`

`if <logic_expr> then if <logic_expr> then <stmt> else <stmt>`



# Extended BNF

- [ ] : optional parts (0 or 1)
- ( | | ) : alternative parts of RHSs.
- { } : repetitions (0 or more). { }+ represents 1 or more.

## BNF

```
<if_stmt> → if (<expr>) <stmt>  
          | if (<expr>) <stmt>  
            else <stmt>
```

```
<expr> → <expr> + <term>  
        | <expr> - <term>  
        | <term>
```

```
<compound> → begin <stmts> end  
<stmts> → <stmts> <stmt>  
          | <stmt>
```

## Extended BNF

```
<if_stmt> → if (<expr>) <stmt>  
           [else <stmt>]
```

```
<expr> → <term> {(+|-) <term>}
```

```
<compound> → begin {<stmt>}+ end
```



# Extended BNF

---

- Recent Variations:

- Alternative RHSs are put on separate lines (instead of using | ).
- Use of a colon(:) instead of  $\rightarrow$ .
- Use of `opt` for optional parts.

`ConstructorDeclare`  $\rightarrow$  `SimpleName` ( `FormalParamListopt` )

- Use of `one of` for choices.

`AssignmentOperator`  $\rightarrow$  one of `=` `*=` `/=` `%=` `+=` `-=` `<<=` `>>=` ...

# Grammars and Recognizers

---

- Recognizer for a given context-free grammar can be constructed algorithmically.
  - yacc (yet another compiler compiler).

# Static Semantics

---

- Nothing to do with ‘meaning’.
- Context-free grammars (CFGs) cannot describe all of the syntax of programming languages.
- Categories of constructs that are trouble:
  - Context-free, but cumbersome.  
(e.g., types of operands in expressions)
  - Non-context-free.  
(e.g., variables must be declared before they are used)

# Attribute Grammars

---

- Attribute grammars (AGs) have additions to CFGs to carry some semantic info on parse tree nodes.
  - Attributes (to symbols).
  - Attribute computation functions (semantic functions) (to rules)
  - Predicate functions
- Primary value of AGs:
  - Static semantics specification.
  - Compiler design (static semantics checking).

# Attribute Grammars: Definition

---

- Def: An **attribute grammar** is a context-free grammar with the following additions:
  - For each grammar symbol  $x$  there is a set  $A(x)$  of attribute values.
    - $A(x) = S(x) \cup I(x)$ ; synthesized and inherited attributes.
  - Each rule  $x_0 \rightarrow x_1 \dots x_n$  has a set of functions that define certain attributes of the nonterminals in the rule.
    - $S(x_0) = f(A(x_1), \dots, A(x_n))$ .
    - $I(x_j) = f(A(x_0), \dots, A(x_n))$ , for  $1 \leq j \leq n$ .
  - Each rule has a (possibly empty) set of predicates to check for attribute consistency.
    - Boolean expression on  $\{A(x_0), \dots, A(x_n)\}$ .
    - False predicate function value: violation of the syntax or static semantic rules.

# Attribute Grammars: Definition

---

- **Intrinsic attributes** on the leaf nodes.
  - e.g. type of a variable comes from the symbol table, which is set from an earlier declaration statement.  

```
int i;  
...  
i = i + 10;
```
- The parse tree is said to be **fully attributed** if all the attribute values are computed.

# Attribute Grammars: An Example

## An attribute grammar example

Syntax rule:	<proc_def> → <b>procedure</b> <proc_name>[1] <proc_body> <b>end</b> <proc_name>[2] ;
Predicate:	<proc_name>[1].string == <proc_name>[2].string

### Example 1:

```
procedure MyFunction
    ...
end MyFunction;
```

Example 2:

```
procedure MyFunction1
    ...
end MyFunction2;
```

# Attribute Grammars: Simple Assignment

An attribute grammar for simple assignment

Syntax rule:	$\langle \text{assign} \rangle \rightarrow \langle \text{var} \rangle = \langle \text{expr} \rangle$
	$\langle \text{expr} \rangle \rightarrow \langle \text{var} \rangle + \langle \text{var} \rangle \mid \langle \text{var} \rangle$
	$\langle \text{var} \rangle \rightarrow A \mid B \mid C$


- Example attribute grammar for type checking:
  - actual\_type: a synthesized attribute for  $\langle \text{var} \rangle$  and  $\langle \text{expr} \rangle$ .
  - expected\_type: an inherited attribute for  $\langle \text{expr} \rangle$ .



# Attribute Grammars: Simple Assignment

**Figure 3.6**

A parse tree for  
 $A = A + B$



# Attribute Grammars: Simple Assignment

An attribute grammar for simple assignment


Syntax rule:	<code>&lt;assign&gt; → &lt;var&gt; = &lt;expr&gt;</code>
Semantic rule:	<code>&lt;expr&gt;.expected_type ← &lt;var&gt;.actual_type</code>
Syntax rule:	<code>&lt;expr&gt; → &lt;var&gt;[2] + &lt;var&gt;[3]</code>
Semantic rule:	<code>&lt;expr&gt;.actual_type ←     if (var[2].actual_type == int) and         (var[3].actual_type == int)     then int     else real endif</code>
Predicate:	<code>&lt;expr&gt;.actual_type == &lt;expr&gt;.expected_type</code>
Syntax rule:	<code>&lt;expr&gt; → &lt;var&gt;</code>
Semantic rule:	<code>&lt;expr&gt;.actual_type ← &lt;var&gt;.actual_type</code>
Predicate:	<code>&lt;expr&gt;.actual_type == &lt;expr&gt;.expected_type</code>
Syntax rule:	<code>&lt;var&gt; → A   B   C</code>
Semantic rule:	<code>&lt;var&gt;.actual_type ← look_up(&lt;var&gt;.string)</code>

# Attribute Grammars (continued)

---


- How are attribute values computed?
  - Inherited attributes: decorated in top-down order.
  - Synthesized attributes: decorated in bottom-up order.
  - In many cases, both kinds of attributes are used, and it is some combination of top-down and bottom-up that must be used.

# Attribute Grammars: Simple Assignment



# Attribute Grammars: Simple Assignment

$\langle \text{var} \rangle.\text{actual\_type} \leftarrow \text{look\_up}(A)$   
 $\langle \text{expr} \rangle.\text{expected\_type} \leftarrow \langle \text{var} \rangle.\text{actual\_type}$   
 $\langle \text{var} \rangle[2].\text{actual\_type} \leftarrow \text{look\_up}(A)$   
 $\langle \text{var} \rangle[3].\text{actual\_type} \leftarrow \text{look\_up}(B)$   
 $\langle \text{expr} \rangle.\text{actual\_type} \leftarrow \text{int or real}$   
 $\langle \text{expr} \rangle.\text{expected\_type} == \langle \text{expr} \rangle.\text{actual\_type}$



What happens if `A.actual_type` is `int_type` and `B.actual_type` is `real_type`?

# (Dynamic) Semantics

---

- There is no single widely acceptable notation or formalism for describing semantics.
- Needs for a methodology and notation for semantics:
  - Programmers need to know what statements mean.
  - Compiler writers must know exactly what language constructs do.
  - Correctness proofs would be possible.
  - Compiler generators would be possible.
  - Designers could detect ambiguities and inconsistencies.

# Operational Semantics

---

- Operational Semantics
  - Describe the meaning of a program by executing its statements on a machine, either simulated or actual.
  - The change in the state of the machine (memory, registers, etc.) defines the meaning of the statement.
  - **Natural** operational semantics : the final result of a program.
  - **Structural** operational semantics : the precise meaning of a program.
- To use operational semantics for a high-level language:
  - Design an appropriate intermediate language - clarity.
  - Virtual machine for the intermediate language is needed in natural operational semantics.

# Operational Semantics

- Example:

C Statement	Meaning
<pre><b>for</b> (expr1; expr2; expr3) {     ... }</pre>	<pre>    expr1; loop: <b>if</b> expr == 0 <b>goto</b> out     ...     expr3;     <b>goto</b> loop out:</pre>

- Uses of operational semantics:
  - Language manuals and textbooks.
  - Teaching programming languages.
- Evaluation
  - Good if used informally (language manuals, etc.).
  - Extremely complex if used formally (e.g., VDL).



# Denotational Semantics

---


- The most rigorous and widely known method.
  - Based on recursive function theory.
  - Originally developed by Scott and Strachey (1970).
  - The meaning of language constructs are defined by only the values of the program's variables.
  - Operational semantics: simpler language vs. mathematical objects.
- Building a denotational specification for a language:
  - Define a mathematical object for each language entity.
  - Define a function that maps instances of the language entities onto instances of the corresponding mathematical objects.
    - Syntactic domain  $\rightarrow$  semantic domain.

# Denotational Semantics: Examples

- Binary numbers.

Binary numbers

```
<bin_num> → 0 | 1
           | <bin_num> 0
           | <bin_num> 1
```



- E.g. Parse tree for '110':

- Syntactic domain: all string representations of binary numbers.
- Semantic domain: non-negative decimal numbers.
- Semantic functions:

$\text{Mbin}('0') = 0$

$\text{Mbin}('1') = 1$

$\text{Mbin}(\langle \text{bin\_num} \rangle '0') = 2 * \text{Mbin}(\langle \text{bin\_num} \rangle)$

$\text{Mbin}(\langle \text{bin\_num} \rangle '1') = 2 * \text{Mbin}(\langle \text{bin\_num} \rangle) + 1$

# Denotational Semantics: Examples

- Decimal numbers.

Decimal numbers

<dec_num>	→	0		1		2		3	...		9
				<dec_num>		0					
				<dec_num>		1					
				...							
				<dec_num>		9					

- E.g. '352'

- Syntactic domain: all string representations of decimal numbers.
- Semantic domain: non-negative decimal numbers.
- Semantic functions:

$Mdec('0') = 0, Mdec('1') = 1, Mdec('2') = 2, \dots,$

$Mdec('9') = 9$

$Mdec(<dec\_num> '0') = 10 * Mdec(<dec\_num>)$

$Mdec(<dec\_num> '1') = 10 * Mdec(<dec\_num>) + 1$

$\dots$

$Mdec(<dec\_num> '9') = 10 * Mdec(<dec\_num>) + 9$

# State of Program

---

- The state of a program is the values of all its current variables.

$$s = \{ \langle i_1, v_1 \rangle, \langle i_2, v_2 \rangle, \dots, \langle i_n, v_n \rangle \}$$

- $i_j$  : the name of a variable,  $v_j$  : the current value of the variable.
- `undef` represents the value is currently undefined.
- $\text{VARMAP}(i_j, s) \rightarrow v_j$  :

map from states to states (or to values for expressions).

- Example:

$$s = \{ \langle 'a', 1 \rangle, \langle 'b', 2 \rangle, \langle 'c', 3 \rangle \}$$
$$\text{VARMAP}('a', s) \rightarrow 1$$
$$\text{VARMAP}('i', s) \rightarrow \text{undef}$$

# Expressions

## Expressions

```
<expr> → <var> | <dec_num> | <binary_expr>
<binary_expr> → <left_expr> <operator> <right_expr>
<left_expr> → <dec_num> | <var>
<right_expr> → <dec_num> | <var>
<operator> → + | *
```

- Map expressions onto  $\mathbb{Z} \cup \{\text{error}\}$ .
  - We assume expressions have no side effects.
  - We assume expressions are decimal numbers, variables, or binary expressions having one arithmetic operator and two operands, each of which can be a variable or a decimal number.

# Expressions

```
Me(<expr>, s) Δ=  
  case <expr> of  
    <dec_num> => Mdec(<dec_num>, s)  
    <var> =>  
      if (VARMAP(<var>, s) == undef)  
      then error  
      else VARMAP(<var>, s)  
    <binary_expr> =>  
      if (Me(<binary_expr>.<left_expr>, s) == error OR  
          Me(<binary_expr>.<right_expr>, s) == error)  
      then error  
      else  
        if (<binary_expr>.<operator> == '+')  
        then Me(<binary_expr>.<left_expr>, s) +  
              Me(<binary_expr>.<right_expr>, s)  
        else Me(<binary_expr>.<left_expr>, s) *  
              Me(<binary_expr>.<right_expr>, s)
```

# Assignment Statements

- Maps state sets to state sets  $\cup \{\text{error}\}$

```
 $M_a(x = E, s) \Delta =$   
  if ( $M_e(E, s) == \text{error}$ )  
  then error  
  else  
     $s' = \{ \langle i_1, v_1' \rangle, \langle i_2, v_2' \rangle, \dots, \langle i_n, v_n' \rangle \}, \text{ where}$   
    for  $j = 1, 2, \dots, n,$   
      if ( $i_j == x$ )  
      then  $v_j' = M_e(E, s)$   
      else  $v_j' = \text{VARMAP}(i_j, s)$ 
```

Example:  $a = b + c, \{ \langle a, 0 \rangle, \langle b, 1 \rangle, \langle c, 2 \rangle \}$

```
 $M_a(a = b + c, \{ \langle a, 0 \rangle, \langle b, 1 \rangle, \langle c, 2 \rangle \})$   
   $M_e(b + c, \{ \langle a, 0 \rangle, \langle b, 1 \rangle, \langle c, 2 \rangle \}) = 3$   
 $\rightarrow \{ \langle a, 3 \rangle, \langle b, 1 \rangle, \langle c, 2 \rangle \}$ 
```

# Logical Pretest Loops

---

- Maps state sets to state sets  $U \{\text{error}\}$ 
  - $M_b$  maps Boolean expressions to Boolean values (or error).
  - $M_{s1}$  maps statement lists and states to states (or error).

```
M1(while B do L, s) Δ=  
  if (Mb(B, s) == undef)  
  then error  
  else if (Mb(B, s) == false)  
    then s  
    else if (Ms1(L, s) == error)  
      then error  
      else M1(while B do L, Ms1(L, s))
```



# Logical Pretest Loops

---

- The loop has been converted from iteration to recursion.
  - The recursive control is mathematically defined by other recursive state mapping functions.
  - Recursion is easier to describe with mathematical rigor.

# Logical Pretest Loop Example

```
M1(while B do L, s) Δ=  
  if (Mb(B, s) == undef)  
  then error  
  else if (Mb(B, s) == false)  
    then s  
    else if (Ms1(L, s) == error)  
      then error  
      else M1(while B do L, Ms1(L, s))
```

Example: **while** i > 0 **do** sum = sum + i; i--;

```
M1(while i > 0 do a = a + i; --i, { <a,0>, <i,3> })  
→ M1(while i > 0 do a = a + i; --i, { <a,3>, <i,2> })  
→ M1(while i > 0 do a = a + i; --i, { <a,5>, <i,1> })  
→ M1(while i > 0 do a = a + i; --i, { <a,6>, <i,0> })  
→ { <a,6>, <i,0> }
```

# Denotational Semantics

---

- Evaluation of denotational semantics:
  - Can be used to prove the correctness of programs.
  - Provides a rigorous way to think about programs.
  - Can be an aid to language design.
    - Revise the design if it is too complex and difficult.
  - Has been used in compiler generation systems.
  - Because of its complexity, it is of little use to language users.

# Axiomatic Semantics

---

- Axiomatic semantics:
  - Based on mathematical logic (predicate calculus).
    - What can be proven about the program?
  - Original purpose: formal program verification.
    - Also used for program semantics specification.
  - Axioms or inference rules are defined for each statement type in the language (to allow transformations of logic expressions into more formal logic expressions).

# Axiomatic Semantics

---

- The logic expressions are called assertions (predicates).
  - **Precondition:** assertion before a statement, stating the relationships and constraints among variables that are true at that point in execution.
  - **Postcondition:** assertion following a statement.
  - The precondition of a statement is the postcondition of the previous statement.
  - Preconditions for the statements are computed from given postconditions.
- Pre-, post form:  $\{P\}$  statement  $\{Q\}$ 
  - Example:  $\text{sum} = 2 * x + 1 \quad \{ \text{sum} > 1 \}$
  - What would be the possible precondition for the postcondition?

# Weakest Precondition


---

- A **weakest precondition** is the least restrictive precondition that will guarantee the postcondition.
  - Example:  $\text{sum} = 2 * x + 1 \quad \{ \text{sum} > 1 \}$   
 $\{ x > 10 \}, \{ x > 50 \}, \{ x > 0 \}, \dots$
- Program proof process:
  - The postcondition for the entire program is the desired result.
  - Work back through the program to the first statement.  
If the precondition on the first statement is the same as the program specification, the program is correct.

# Inference Rule

---

- Inference rule:



- If  $s_1, s_2, \dots$ , and  $s_n$  are true, then the truth of  $s$  can be inferred.
- **Axiom**: a logical statement that is assumed to be true.
- Either an axiom or an inference rule must be available for each kind of statement in the language.

# Assignment Statements

---

- Axiom for assignment statements:

$$\{Q_{x \rightarrow E}\} \ x = E \ \{Q\}$$

- $Q_{x \rightarrow E}$  :  $Q$  with all instances of  $x$  replaced by  $E$ .
- Examples:

$a = b / 2 - 1 \ \{ a < 10 \}$   
Weakest precondition:  $b < 22$

$x = 2 * y - 3 \ \{ x > 25 \}$   
Weakest precondition:  $y > 14$

$x = x + y - 3 \ \{ x > 10 \}$   
Weakest precondition:  $y > 13 - x$



# Assignment Statements

---

- Consider the following logical statement.

$$\{ x > 3 \} \quad x = x - 3 \quad \{ x > 0 \}$$

- Can we prove this statement?

$$x = x - 3 \quad \{ x > 0 \}$$

Weakest precondition:  $x > 3$

- This is same as the precondition – proven!

# Assignment Statements

- Another example:

$\{ x > 5 \} \quad x = x - 3 \quad \{ x > 0 \}$

Weakest precondition:  $x > 3$

- However  $\{ x > 5 \}$  implies  $\{ x > 3 \}$ .

- Rule of consequence:

$$\frac{\{ P \} \ S \ \{ Q \}, \quad P' \Rightarrow P, \quad Q \Rightarrow Q'}{\{ P' \} \ S \ \{ Q' \}}$$

- Precondition can always be strengthened.
- Postcondition can always be weakened.

$$\frac{\{ x > 3 \} \ x = x - 3 \ \{ x > 0 \}, \quad \{ x > 5 \} \Rightarrow \{ x > 3 \}, \quad \{ x > 0 \} \Rightarrow \{ x > 0 \}}{\{ x > 5 \} \ x = x - 3 \ \{ x > 0 \}}$$

# Sequences

- An inference rule for sequences of the form  $s1; s2$

$$\begin{array}{l} \{ P1 \} \quad S1 \quad \{ P2 \} \\ \{ P2 \} \quad S2 \quad \{ P3 \} \end{array}$$

$$\frac{\{ P1 \} \quad S1 \quad \{ P2 \}, \quad \{ P2 \} \quad S2 \quad \{ P3 \}}{\{ P1 \} \quad S1; S2 \quad \{ P3 \}}$$

- Sequences of assignments:

$$x1 = E1; x2 = E2$$

$$\{ P_{x2 \rightarrow E2} \} \quad x2 = E2 \quad \{ P \}$$

$$\{ (P_{x2 \rightarrow E2})_{x1 \rightarrow E1} \} \quad x1 = E1 \quad \{ P_{x2 \rightarrow E2} \}$$

- Weakest precondition with postcondition  $P$  is  $\{ (P_{x2 \rightarrow E2})_{x1 \rightarrow E1} \}$ .
- Example:

$$\begin{array}{l} y = 3 * x + 1; x = y + 3; \quad \{ x < 10 \} \\ \{ y < 7 \} \quad x = y + 3; \quad \{ x < 10 \} \\ \{ x < 2 \} \quad y = 3 * x + 1; x = y + 3; \quad \{ x < 10 \} \end{array}$$

# Selection

- An inference rule for selection statements:

$$\frac{\{B \text{ and } P\} S1 \{Q\}, \{(\text{not } B) \text{ and } P\} S2 \{Q\}}{\{P\} \text{ **if** } B \text{ **then** } S1 \text{ **else** } S2 \{Q\}}$$

- Example:

**if**  $x > 0$  **then**  $y = y - 1$  **else**  $y = y + 1$   $\{ y > 0 \}$

- $\{ y > 1 \}$  is the w.p. of  $y = y - 1$   $\{ y > 0 \}$
- $\{ y > -1 \}$  is the w.p. of  $y = y + 1$   $\{ y > 0 \}$
- Therefore,

$\{ y > 1 \}$  **if**  $x > 0$  **then**  $y = y - 1$  **else**  $y = y + 1$   $\{ y > 0 \}$

# Logical Pretest Loops

- An inference rule for logical pretest loops:

$\{P\} \text{ while } B \text{ do } S \text{ end } \{Q\}$

$$\frac{\{I \text{ and } B\} S \{I\}}{\{I\} \text{ while } B \text{ do } S \text{ end } \{I \text{ and } (\text{not } B)\}} \quad *$$

- where  $I$  is the loop invariant.

- How to find the loop invariant?

- $P \Rightarrow I$  <sup>①</sup>: the loop invariant must be true initially.
- $\{I \text{ and } B\} S \{I\}$  <sup>②</sup>:  $I$  is not changed by executing the loop body.
- $(I \text{ and } (\text{not } B)) \Rightarrow Q$  <sup>③</sup>: if  $I$  is true and  $B$  is false,  $Q$  is implied.
- The loop terminates. <sup>④</sup>: this can be difficult to prove.

$$\frac{\{I \text{ and } B\} S \{I\}}{\{I\} \text{ while } B \text{ do } S \text{ end } \{I \text{ and } (\text{not } B)\}} \quad *, \quad P \Rightarrow I$$

$\{P\} \text{ while } B \text{ do } S \text{ end } \{Q\}$

# Logical Pretest Loop: Example

- Weakest precondition predicate transformer:

$\text{wp}(\text{statement}, \text{postcondition}) = \text{precondition}$

- It takes a predicate and returns the weakest precondition of the statement which is another predicate.

- Example:

```
while y <> x do y = y + 1 end { y == x }  
  begin: { y == x }  
  1 iter: wp(y = y + 1, { y == x }) = { y == x-1 }  
  2 iter: wp(y = y + 1, { y == x-1 }) = { y == x-2 }  
  3 iter: wp(y = y + 1, { y == x-2 }) = { y == x-3 }  
        ...  
        { y <= x }  
{ y <= x } while y <> x do y = y + 1 end { y == x }
```

# Logical Pretest Loop

- $\{P\} \text{ while } B \text{ do } S \text{ end } \{Q\}$

Example:  $\{ y \leq x \} \text{ while } y \neq x \text{ do } y = y + 1 \text{ end } \{ y = x \}$

- Does the invariant  $I = \{ y \leq x \}$  satisfy the four criteria?
  - $P \Rightarrow I$  : true since  $P = I$ .
  - $\{I \text{ and } B\} S \{I\}$ 
    - $\{I \text{ and } B\} : \{ y \leq x \text{ and } y \neq x \}$
    - Using assignment axiom:  $\{ y < x \}$  is the w.p. of  $y = y + 1 \{ y \leq x \}$
    - True since  $\{ y \leq x \text{ and } y \neq x \} \Rightarrow \{ y < x \}$
  - $(I \text{ and } (\text{not } B)) \Rightarrow Q$ 
    - $\{ (y \leq x) \text{ and not } (y \neq x) \} \Rightarrow \{ y = x \}$
    - $\{ (y \leq x) \text{ and } (y = x) \} \Rightarrow \{ y = x \}$
  - The loop terminates.
    - For integer  $x$  and  $y$ , the loop terminates.

# Logical Pretest Loop: Another Example

- **while**  $s > 1$  **do**  $s = s / 2$  **end** {  $s == 1$  }  
     $\text{wp}(s = s / 2, \{ s == 1 \}) = \{ s / 2 == 1 \}, \text{ or } \{ s == 2 \}$   
     $\text{wp}(s = s / 2, \{ s == 2 \}) = \{ s / 2 == 2 \}, \text{ or } \{ s == 4 \}$   
    ...  
    {  $s == 2^i, i \geq 0$  } **while**  $s > 1$  **do**  $s = s / 2$  **end** {  $s == 1$  }  
  
- This is not the weakest precondition (for integer operations).
- {  $s > 1$  } **while**  $s > 1$  **do**  $s = s / 2$  **end** {  $s == 1$  }



# Loop Invariant

```
{P} while B do S end {Q}
```

$P \Rightarrow I$

$\{I \text{ and } B\} S \{I\}$

$(I \text{ and } (\text{not } B)) \Rightarrow Q$

The loop terminates.

$\{I \text{ and } B\} S \{I\}$

---

$\{I\} \text{ while } B \text{ do } S \text{ end } \{I \text{ and } (\text{not } B)\}$

- The loop invariant  $I$ :
  - A weakened version of the loop postcondition, and also a precondition.
  - Weak enough to be satisfied prior to the beginning of the loop.
  - Strong enough to force the truth of the postcondition, when combined with the loop exit condition.
- Loop termination is hard to prove.
  - Total correctness vs. partial correctness.

# Program Proofs: Example 1

---

$\{ x == A \text{ and } y == B \}$

$t = x; x = y; y = t; \quad \{ x == B \text{ and } y == A \}$

- Use assignment axiom.

$\{ x == B \text{ and } t == A \} \quad y = t; \quad \{ x == B \text{ and } y == A \}$

$\{ y == B \text{ and } t == A \}$

$x = y; y = t; \quad \{ x == B \text{ and } y == A \}$

$\{ y == B \text{ and } x == A \}$

$t = x; x = y; y = t; \quad \{ x == B \text{ and } y == A \}$

# Program Proofs: Example 2

---

```
{ n >= 0 }  
  count = n;  fact = 1;  
  while count <> 0 do  
    fact = fact * count;  
    count = count - 1;  
  end  
{ fact == n! }
```

- Find the loop invariant (using some ingenuity!):

$$\text{fact} == (\text{count}+1) * (\text{count}+2) * \dots * (n-1) * n == n! / \text{count}!$$
$$I = (\text{fact} == n! / \text{count}!) \text{ and } (\text{count} \geq 0)$$

- Check the four criteria for the loop.

# Program Proofs: Ex

```
{ n >= 0 }
  count = n;  fact = 1;
  while count <> 0 do
    fact = fact * count;
    count = count - 1;
  end
{ fact == n! }
```

- Loop body:

- $P \Rightarrow I$

It holds since  $P == I$ .  $I = (fact == n!/count!) \text{ and } (count \geq 0)$

- $\{I \text{ and } B\} S \{I\}$

$\{I \text{ and } B\} :$

$\{(fact == n!/count!) \text{ and } (count \geq 0) \text{ and } (count \neq 0)\}$

$\{I\} : \{(fact == n!/count!) \text{ and } (count \geq 0)\}$

$\{P_c\} \text{ count} = \text{count} - 1; \{I\}$

$P_c = (fact == n!/(count-1)!) \text{ and } (count \geq 1)$

$\{P'\} \text{ fact} = \text{fact} * \text{count}; \{P_c\}$

$P' = (fact == n!/count!) \text{ and } (count \geq 1)$

It holds since  $I \text{ and } B \Rightarrow P'$ .

# Program Proofs: Ex

```
{ n >= 0 }
[ count = n; fact = 1;
  while count <> 0 do
    fact = fact * count;
    count = count - 1;
  end
{ fact == n! }
```

$I = (fact == n!/count!) \text{ and } (count \geq 0)$

- Loop body (cont.):

-  $(I \text{ and } (\text{not } B)) \Rightarrow Q$

$((fact == n!/count!) \text{ and } (count \geq 0) \text{ and not } (count \neq 0))$   
 $\Rightarrow ((fact == n!/count!) \text{ and } (count == 0))$  \*\*note  $0! = 1$   
 $\Rightarrow fact == n!$

- Entire program:

$\{Pi2\} \text{ fact} = 1; \{P\}$

$\{Pi2\} \text{ fact} = 1; \{(fact == n!/count!) \text{ and } (count \geq 0)\}$   
 $\Rightarrow Pi2 = ((1 == n!/count!) \text{ and } (count \geq 0))$

$\{Pi1\} \text{ count} = n; \{Pi2\}$

$\{Pi1\} \text{ count} = n; \{(1 == n!/count!) \text{ and } (count \geq 0)\}$   
 $\Rightarrow Pi1 = (1 == n!/n! \text{ and } (n \geq 0))$   
 $= (n \geq 0)$

# Evaluation of Axiomatic Semantics

---

- Evaluation of axiomatic semantics:
  - Developing axioms or inference rules for all of the statements in a language is difficult.
  - It is a good tool for correctness proofs, and an excellent framework for reasoning about programs, but it is not as useful for language users and compiler writers.
  - Its usefulness in describing the meaning of a programming language is limited for language users or compiler writers.

# Summary

---

- BNF and context-free grammars are equivalent meta-languages.
  - Well-suited for describing the syntax of programming languages.
- An attribute grammar is a descriptive formalism that can describe both the syntax and the semantics of a language.
- Three primary methods of semantics description
  - Operation, denotational, axiomatic.